

Guarana

A GAP4 Package

Version 0.92

by

Björn Assmann

Mathematical Institute

University of St Andrews

email: bjorn@mcs.st-and.ac.uk

June 2007

Contents

1	Introduction	3
1.1	Setup for computing the correspondence	3
1.2	Collection	3
2	Computing the Mal'cev correspondence	4
2.1	The main functions	4
2.2	An example application	5
3	Mal'cev collection	7
3.1	The main functions	7
3.2	An example application	8
4	Installation	9
4.1	Installing this package	9
4.2	Loading the Guarana package	9
4.3	Running the test suite	9
	Bibliography	10
	Index	11

1

Introduction

In this package we demonstrate the algorithmic usefulness of the so-called Mal'cev correspondence for computations with infinite polycyclic groups; it is a correspondence that associates to every \mathbb{Q} -powered nilpotent group H a unique rational nilpotent Lie algebra L_H and vice-versa. The Mal'cev correspondence was discovered by Anatoly Mal'cev in 1951 [Mal51].

1.1 Setup for computing the correspondence

Let G be a finitely generated torsion-free nilpotent group, i.e. a T -group. Then G can be embedded in a \mathbb{Q} -powered hull \hat{G} . The group \hat{G} is a \mathbb{Q} -powered nilpotent group and is unique up to isomorphism. We denote the Lie algebra which corresponds to \hat{G} under the Mal'cev correspondence by $L(G) = L_{\hat{G}}$.

We provide an algorithm for setting up the Mal'cev correspondence between \hat{G} and the Lie algebra $L(G)$. That is, if G is given by a polycyclic presentation with respect to a Mal'cev basis, then we can compute a structure constants table of $L(G)$. Furthermore for a given $g \in G$ we can compute the corresponding element in $L(G)$ and vice versa.

1.2 Collection

Every element of a polycyclically presented group has a unique normal form. An algorithm for computing this normal form is called a collection algorithm. Such an algorithm lies at the heart of most methods dealing with polycyclically presented groups. The current state of the art is collection from the left [Geb02,LGS90,VLe90].

This package contains a new collection algorithm for polycyclically presented groups, which we call Mal'cev collection [ALi07]. Mal'cev collection is in some cases dramatically faster than collection from the left, while using less memory.

2 Computing the Mal'cev correspondence

2.1 The main functions

Let G be a T -group and \hat{G} its \mathbb{Q} -powered hull. In this chapter we describe functionality for setting up the Mal'cev correspondence between \hat{G} and the Lie algebra $L(G)$.

The data structures needed for computations with \hat{G} and $L(G)$ are stored in a so-called Mal'cev object. Computational representations of elements of \hat{G} , respectively $L(G)$, will be called Mal'cev group elements, respectively Mal'cev Lie elements.

1 ▶ `MalcevObjectByTGroup(N)`

If N is a T -group (i.e. a finitely generated torsion-free nilpotent group), given by a polycyclic presentation with respect to a Mal'cev basis, then this function computes the Mal'cev correspondence for N and stores the result in a so-called Mal'cev object. Otherwise this function returns `fail`. In the moment this function is restricted to groups N of nilpotency class at most 9.

2 ▶ `UnderlyingGroup(mo)`

For a Mal'cev object mo this function returns the T -group, which was used to build mo .

3 ▶ `UnderlyingLieAlgebra(mo)`

For a Mal'cev object mo this function returns the Lie algebra, which underlies the correspondence described by mo .

4 ▶ `Dimension(mo)`

returns the dimension of the Lie algebra that underlies the Mal'cev object mo .

5 ▶ `MalcevGrpElementByExponents(mo, exps)`

For a Mal'cev object mo and an exponent vector $exps$ with rational entries, this function returns the Mal'cev group element, which has exponents $exps$ with respect to the Mal'cev basis of the underlying group of mo .

6 ▶ `MalcevLieElementByCoefficients(mo, coeffs)`

For a Mal'cev object mo and a coefficient vector $coeffs$ with rational entries, this function returns the Mal'cev Lie element, which has coefficients $coeffs$ with respect to the basis of the underlying Lie algebra of mo .

7 ▶ `RandomGrpElm(mo, range)`

For a Mal'cev object mo this function returns the output of `MalcevGrpElementByExponents(mo, exps)`, where $exps$ is an exponent vector whose entries are randomly chosen integers between $-range$ and $range$.

8 ▶ `RandomLieElm(mo, range)`

For a Mal'cev object mo this function returns the output of `MalcevLieElementByExponents(mo, coeffs)`, where $coeffs$ is a coefficient vector whose entries are randomly chosen integers between $-range$ and $range$.

- 9 ▶ `Log(g)`
For Mal'cev group element g this function returns the corresponding Mal'cev Lie element.
- 10 ▶ `Exp(x)`
For Mal'cev Lie element x this function returns the corresponding Mal'cev group element.
- 11 ▶ `g * h`
returns the product of Mal'cev group elements.
- 12 ▶ `Comm(x, y)`
If x, y are Mal'cev group elements, then this function returns the group theoretic commutator of x and y . If x, y are Mal'cev Lie elements, then this functions returns the Lie commutator of x and y .
- 13 ▶ `MalcevSymbolicGrpElementByExponents(mo, expts)`
For a Mal'cev object mo and an exponent vector $expts$ with rational indeterminates as entries, this functions returns the Mal'cev group element, which has exponents $expts$ with respect to the Mal'cev basis of the underlying group of mo .
- 14 ▶ `MalcevLieElementByCoefficients(mo, coeffs)`
For a Mal'cev object mo and a coefficient vector $coeffs$ with rational indeterminates as entries, this functions returns the Mal'cev Lie element, which has coefficients $coeffs$ with respect to the basis of the underlying Lie algebra of mo .

2.2 An example application

```

gap> n := 2;
2
gap> F := FreeGroup( n );
<free group on the generators [ f1, f2 ]>
gap> c := 3;
3
gap> N := NilpotentQuotient( F, c );
Pcp-group with orders [ 0, 0, 0, 0, 0 ]

gap> mo := MalcevObjectByTGroup( N );
<<Malcev object of dimension 5>>
gap> dim := Dimension( mo );
5
gap> UnderlyingGroup( mo );
Pcp-group with orders [ 0, 0, 0, 0, 0 ]
gap> UnderlyingLieAlgebra( mo );
<Lie algebra of dimension 5 over Rationals>

gap> g := MalcevGrpElementByExponents( mo, [1,1,0,2,-1/2] );
[ 1, 1, 0, 2, -1/2 ]
gap> x := MalcevLieElementByCoefficients( mo, [1/2, 2, -1, 3, 5] );
[ 1/2, 2, -1, 3, 5 ]

gap> h := RandomGrpElm( mo );
[ 5, -3, 0, -2, 8 ]
gap> y := RandomLieElm( mo );
[ 3, 9, 5, 5, 2 ]

```

```

gap> z := Log( g );
[ 1, 1, -1/2, 7/3, -1/3 ]
gap> Exp( z ) = g;
true
gap> k := Exp( y );
[ 3, 9, 37/2, 77/4, 395/4 ]
gap> Log( k ) = y;
true

gap> g*h;
[ 6, -2, 5, 10, -15/2 ]
gap> Comm(g,h);
[ 0, 0, 8, 10, -18 ]
gap> Comm(x,y);
[ 0, 0, 3/2, -25/4, -79/4 ]

gap> indets := List( List( [1..dim], i->Concatenation( "a_", String(i) ) ),
> x->Indeterminate( Rationals, x : new ) );
[ a_1, a_2, a_3, a_4, a_5 ]
gap> g_sym := MalcevSymbolicGrpElementByExponents( mo, indets );
[ a_1, a_2, a_3, a_4, a_5 ]
gap> x_sym := Log( g_sym );
[ a_1, a_2, -1/2*a_1*a_2+a_3, 1/12*a_1^2*a_2+1/4*a_1*a_2-1/2*a_1*a_3+a_4,
-1/12*a_1*a_2^2+1/4*a_1*a_2-1/2*a_2*a_3+a_5 ]
gap> g_sym * g;
[ a_1+1, a_2+1, a_2+a_3, a_3+a_4+2, 1/2*a_2^2+1/2*a_2+a_3+a_5-1/2 ]

```

3

Mal'cev collection

Let G be an infinite polycyclic group. It is well-known that there exist a normal T -group N and a T -group C such that $H = CN$ is normal of finite index in G and H/N is free abelian of finite rank [Seg83].

In this chapter we present an effective collection method for an infinite polycyclic group which is given by a polycyclic presentation with respect to a polycyclic sequence P going through the normal series $1 \leq N \leq H \leq G$.

This polycyclic sequence P must be chosen as follows. Let (n_1, \dots, n_l) be a Mal'cev basis of N and let (c_1N, \dots, c_kN) be a basis for the free abelian group CN/N . Then $(c_1, \dots, c_k, n_1, \dots, n_l)$ is a polycyclic sequence for $H = CN$. Further there exists $f_1, \dots, f_j \in G$ such that (f_1H, \dots, f_jH) is a polycyclic sequence for G/H . Now we set

$$P = (f_1, \dots, f_j, c_1, \dots, c_k, n_1, \dots, n_l)$$

3.1 The main functions

1 ► `MalcevCollectorConstruction([G, inds, C, CC, N, NN])`

returns a Mal'cev collector for the infinite polycyclically presented group G . The group G must be given with respect to a polycyclic sequence $(g_1, \dots, g_r, c_{r+1}, \dots, c_{r+s}, n_{r+s+1}, \dots, n_{r+s+t})$ with the following properties:

- (a) $(n_{r+s+1}, \dots, n_{r+s+t})$ is a Mal'cev basis for the T -group $N \leq G$,
- (b) $(c_{r+1}N, \dots, c_{r+s}N)$ is a basis for the free-abelian group CN/N where $C \leq G$ is a T -group generated by c_{r+1}, \dots, c_{r+s} ,
- (c) (g_1CN, \dots, g_rCN) is a polycyclic sequence for the finite group G/CN .

The list `inds` is equal to $[[1, \dots, r], [r+1, \dots, r+s], [r+s+1, \dots, r+s+t]]$. The group CC is isomorphic to C via `CC!.bijection` and given by a polycyclic presentation with respect to a Mal'cev basis starting with c_{r+1}, \dots, c_{r+s} . The group NN is isomorphic to N via `NN!.bijection`. and given by a polycyclic presentation with respect to the Mal'cev basis $(n_{r+s+1}, \dots, n_{r+s+t})$.

2 ► `GUARANA.Tr_n_O1(n)`
► `GUARANA.Tr_n_O2(n)`

for a positive integer n these functions construct polycyclically presented groups that can be used to test the Mal'cev collector. They return a list which can be used as input for the function `MalcevCollectorConstruction`. The constructed groups are isomorphic to triangular matrix groups of dimension n over the ring O_1 , respectively O_2 . The ring O_1 , respectively O_2 , is the maximal order of $\mathbb{Q}(\theta_i)$ where θ_1 , respectively θ_2 , is a zero of the polynomial $p_1(x) = x^2 - 3$, respectively $p_2(x) = x^3 - x^2 + 4$.

3 ► `GUARANA.F_2c_Aut1(c)`
► `GUARANA.F_3c_Aut2(c)`

for a positive integer c these functions construct polycyclically presented groups that can be used to test the Mal'cev collector. They return a list which can be used as input for the function `MalcevCollectorConstruction`. These groups are constructed as follows: Let $F_{n,c}$ be the free nilpotent of class c group on n generators. An automorphism φ of the free group F_n naturally induces an automorphism $\bar{\varphi}$ of $F_{n,c}$. We use

the automorphism φ_1 of F_2 which maps f_1 to f_2^{-1} and f_2 to $f_1 f_2^3$ and the automorphism φ_2 of F_3 mapping f_1 to f_2^{-1} , f_2 to f_3^{-1} and f_3 to $f_2^{-3} f_1^{-1}$ for our construction. The returned group `F_2c_Aut1`, respectively `F_3c_Aut2`, is isomorphic to the semidirect product $\langle \varphi_1 \rangle \rtimes F_{2,c}$, respectively $\langle \varphi_2 \rangle \rtimes F_{3,c}$.

4 ▶ `MalcevGElementByExponents(malCol, exps)`

For a Mal'cev collector `malCol` of a group G and an exponent vector `exps` with integer entries, this function returns the group element of G , which has exponents `exps` with respect to the polycyclic sequence underlying `malCol`.

5 ▶ `Random(malCol, range)`

For a Mal'cev collector `malCol` this function returns the output of `MalcevGElementByExponents(malCol, exps)`, where `exps` is an exponent vector whose entries are randomly chosen integers between `-range` and `range`.

6 ▶ `g * h`

returns the product of group elements which are defined with respect to a Mal'cev collector by the the function `MalcevGElementByExponents`.

7 ▶ `GUARANA.AverageRuntimeCollec(malCol, ranges, no)`

for a Mal'cev collector `malCol`, a list of positive integers `ranges` and a positive integer `no` this function computes for each number r in `ranges` the average runtime of `no` multiplications of two random elements of `malCol` of range r , as generated by `Random(malCol, r)`.

3.2 An example application

```
gap> ll := GUARANA.Tr_n_01( 3 );
[ Pcp-group with orders [ 2, 2, 2, 0, 0, 0, 0, 0, 0, 0, 0 ],
  [ [ 1 .. 3 ], [ 4 .. 6 ], [ 7 .. 12 ] ],
  Pcp-group with orders [ 0, 0, 0, 0, 0, 0 ],
  Pcp-group with orders [ 0, 0, 0, 0, 0, 0 ],
  Pcp-group with orders [ 0, 0, 0 ], Pcp-group with orders [ 0, 0, 0 ] ]
gap> malCol := MalcevCollectorConstruction( ll );
<<Malcev collector>>
  F : [ 2, 2, 2 ]
  C : <<Malcev object of dimension 3>>
  N : <<Malcev object of dimension 6>>

gap> exps_g := [ 1, 1, 1, -3, -2, 1, -2, -1, 0, 3, -1, 3 ];
[ 1, 1, 1, -3, -2, 1, -2, -1, 0, 3, -1, 3 ]
gap> exps_h := [ 1, 0, 1, -1, 0, 2, 0, 4, -1, 5, 9, -5 ];
[ 1, 0, 1, -1, 0, 2, 0, 4, -1, 5, 9, -5 ]
gap> g := MalcevGElementByExponents( malCol, exps_g );
[ 1, 1, 1, -3, -2, 1, -2, -1, 0, 3, -1, 3 ]
gap> h := MalcevGElementByExponents( malCol, exps_h );
[ 1, 0, 1, -1, 0, 2, 0, 4, -1, 5, 9, -5 ]

gap> k := g*h;
[ 0, 1, 0, -4, -2, 3, 1, 4, 35, -16, -404, 232 ]

gap> Random( malCol, 10 );
[ 0, 0, 1, 9, 5, 5, 2, -2, 7, -10, 7, -6 ]
```

4

Installation

4.1 Installing this package

The Guarana package is part of the standard distribution of GAP and so normally there should be no need to install it separately. If by any chance it is not part of your GAP distribution, then the standard method is to unpack the package into the `pkg` directory of your GAP distribution. This will create a `guarana` subdirectory. For other non-standard options please see Chapter 75.1 of the GAP Reference Manual.

Note that the GAP-Packages Polycyclic and Polenta are needed for this package. Normally they should be contained in your distribution. If not, they can be obtained at

www.gap-system.org/Packages/packages.html

4.2 Loading the Guarana package

If the Guarana Package is not already loaded then you have to request it explicitly. This can be done by `LoadPackage("guarana")`. The `LoadPackage` command is described in Section 75.2.1 in the GAP Reference Manual.

4.3 Running the test suite

Once the package is installed, it is possible to check the correct installation by running the test suite of the package.

```
gap> Read(Filename(DirectoriesPackageLibrary("guarana","tst")[1],"testall.g"));
```

For more details on Test Files see Section 7.9 of the GAP Reference Manual.

If the test suite runs into an error, then please send a message to bjoern@mcs.st-and.ac.uk including the error message.

Bibliography

- [ALi06] B. Assmann and S. Linton. Using the Mal'cev correspondence for collection in polycyclic groups. *J. Algebra*, 2007.
- [Eic01] Bettina Eick. Algorithms for polycyclic groups. Habilitationsschrift, Gesamthochschule Kassel, 2001.
- [Geb02] V. Gebhardt. Efficient collection in infinite polycyclic groups. *J. Symbolic Comput.*, 34 (3):213–228, 2002.
- [LGS90] C. R. Leedham-Green and L. H. Soicher. Collection from the left and other strategies. *J. Symbolic Comput.*, 9:665 – 675, 1990.
- [Mal51] A. J. Mal'cev. On certain classes of infinite soluble groups. *Mat. Sb.*, 28:567 – 588, 1951.
- [Seg83] D. Segal. *Polycyclic Groups*. Cambridge University Press, Cambridge, 1983.
- [VLe90] M. Vaughan-Lee. Collection from the left. *J. Symbolic Comput.*, 9:725–733, 1990.

Index

This index covers only this manual. A page number in *italics* refers to a whole section which is devoted to the indexed subject. Keywords are sorted with case and spaces ignored, e.g., “PermutationCharacter” comes before “permutation group”.

A

An example application, *5, 8*

C

Collection, *3*

Comm, *5*

D

Dimension, *4*

E

Exp, *5*

G

Guarana, *3*

GUARANA.AverageRuntimeCollec, *8*

GUARANA.F_2c_Aut1, *7*

GUARANA.F_3c_Aut2, *7*

GUARANA.Tr_n_01, *7*

GUARANA.Tr_n_02, *7*

I

Installation, *9*

L

loading the Guarana package, *9*

Log, *5*

M

MalcevCollectorConstruction, *7*

MalcevGElementByExponents, *8*

MalcevGrpElementByExponents, *4*

MalcevLieElementByCoefficients, *4, 5*

MalcevObjectByTGroup, *4*

MalcevSymbolicGrpElementByExponents, *5*

multiplication, *5, 8*

R

Random, *8*

RandomGrpElm, *4*

RandomLieElm, *4*

Running the test suite, *9*

S

Setup for computing the correspondence, *3*

T

The main functions, *4, 7*

U

UnderlyingGroup, *4*

UnderlyingLieAlgebra, *4*